

miniKanren with fair search strategies

KUANG-CHEN LU, Indiana University

WEIXI MA, Indiana University

DANIEL P. FRIEDMAN, Indiana University

The syntax of a programming language should reflect its semantics. When writing a cond^e expression in miniKanren, a programmer would expect all clauses share the same chance of being explored, as these clauses are written in parallel. The existing search strategy, interleaving depth-first search (DFS_i), however, prioritizes its clauses by the order how they are written down. Similarly, when a cond^e is followed by another goal conjunctively, a programmer would expect states in parallel share the same chance of being explored. Again, the answers by DFS_i is different from the expectation. We have devised three new search strategies that have different level of fairness in disj and conj .

ACM Reference Format:

Kuang-Chen Lu, Weixi Ma, and Daniel P. Friedman. 2019. miniKanren with fair search strategies. 1, 1 (May 2019), 14 pages. <https://doi.org/10.1145/nnnnnnnn.nnnnnnnn>

1 INTRODUCTION

miniKanren is a family of relational programming languages. Friedman et al. [2] introduce miniKanren and its implementation in *The Reasoned Schemer, 2nd Ed* (TR2). miniKanren programs, especially relational interpreters, have been proven to be useful in solving many problems by Byrd et al. [1].

A subtlety arises when a cond^e contains many clauses: not every clause has an equal chance to contribute to the result. As an example, consider the following relation `repeato` and its invocation.

^{c1}LKC: submission deadline: Mon 27 May 2019

^{c2}LKC: `disj` and `conj` occurs free in the abstract.

^{c3}LKC: Shall we delete the sentence about relational interpreters? It is misleading: it is emphasizing relational interpreters, but we say nothing further about them later.

Authors' addresses: Kuang-Chen LuIndiana University; Weixi MaIndiana University; Daniel P. FriedmanIndiana University.

Unpublished working draft. Not for distribution.

Permission to make digital or hard copies of all or part of this work for personal or classroom use is granted without fee provided that copies are not made or distributed for profit or commercial advantage and that copies bear this notice and the full citation on the first page. Copyrights for components of this work owned by others than ACM must be honored. Abstracting with credit is permitted. To copy otherwise, or republish, to post on servers or to redistribute to lists, requires prior specific permission and/or a fee. Request permissions from permissions@acm.org.

© 2019 Association for Computing Machinery.

XXXX-XXXX/2019/5-ART \$15.00

<https://doi.org/10.1145/nnnnnnnn.nnnnnnnn>

```

48      (defrel (repeato x out)
49        (conde
50          [(≡ `(,x) out)]
51          [(fresh (res)
52            (≡ `(,x . ,res) out)
53            (repeato x res))]))
54
55      > (run 4 q
56        (repeato '* q))
57      '((*) (* *) (* * *) (* * * *))

```

Next, consider the following disjunction of invoking repeat^o with four different letters.

```

59      > (run 12 q
60        (conde
61          ((repeato 'a q))
62          ((repeato 'b q))
63          ((repeato 'c q))
64          ((repeato 'd q))))

```

cond^e intuitively relates its clauses with logical or. And thus an unsuspecting beginner would expect each letter to contribute equally to the result, as follows.

```

68      '((a) (b) (c) (d)
69        (a a) (b b) (c c) (d d)
70        (a a a) (b b b) (c c c) (d d d))

```

The cond^e in TR2, however, generates a less expected result.

```

72      '((a) (a a) (b) (a a a)
73        (a a a a) (b b)
74        (a a a a a) (c)
75        (a a a a a a) (b b b)
76        (a a a a a a a) (d))

```

The miniKanren in TR2 implements interleaving DFS (DFS_i), the cause of this unexpected result. With this search strategy, each clause takes half of its received computational resources and pass the other half to its following clauses, except for the last clause that takes all resources it receives. In the example above, the a clause takes half of all recurses. And the b clause takes a quarter. Thus c and d barely contribute to the result.

DFS_i is sometimes powerful for an expert. By carefully organizing the order of cond^e clauses, a miniKanren program can explore more “interesting” clauses than those uninteresting ones, and thus use computational resources efficiently. A little miniKanrener, however, may beg to differ—understanding implementation details and fiddling with clauses order is not the first priority of a beginner.

There is another reason that miniKanren could use more search strategies than just DFS_i. In many applications, there does not exist one order that serves for all purposes. For example, a relational dependent type checker contains clauses for constructors that build data and clauses for eliminators that use data. When the type checker is used to generate simple and shallow programs, the clauses of constructors should be put in the front of cond^e. When performing proof searches for complicated programs, the clauses of eliminator should take the focus. With DFS_i, these two uses cannot be efficient at the same time. In fact, to make one use efficient, the other one must be drastically slow.

The specification that every clause in the same cond^e is given equal “search priority” is called *fair disj*. And search strategies with *almost-fair disj* give every clause similar priority. *Fair conj*, a related concept, is more complicated. We defer it to the next section.

To summarize our contribution, we

- propose and implement balanced interleaving depth-first search (DFS_{bi}), a new search strategy with *almost-fair disj*.
- propose and implement fair depth-first search (DFS_f), a new search strategy with *fair disj*.
- implement in a new way breath-first search (BFS), a search strategy with *fair disj* and *fair conj*. And we prove formally that our BFS implementation is equivalent to the one by Seres et al. [5]. Our code runs faster in all benchmarks and is simpler.

2 SEARCH STRATEGIES AND FAIRNESS

In this section, we define *fair disj*, *almost-fair disj* and *fair conj*. Before going further into fairness, we would like to give a short review about state, search space, and goal, because fairness is defined in terms of them. A *state* is a collection of constraints. Every answer corresponds to a state. A *search space* is a collection of states. And a *goal* is a function from a state to a search space.

Now we elaborate fairness by running more queries about *repeato*.

2.1 fair disj

Given the following program, it is natural to expect lists of each letter to constitute 1/4 in the query result. DFS_i , the current search strategy, however, results in many more lists of *a* than lists of other letters. And some letters (e.g. *c* and *d*) are rarely seen. The situation would be exacerbated if cond^e contains more clauses.

```
;; DFSi (unfair disj)
> (run 12 q
  (conde
    ((repeat0 'a q))
    ((repeat0 'b q))
    ((repeat0 'c q))
    ((repeat0 'd q))))
'((a) (a a) (b) (a a a)
  (a a a a) (b b)
  (a a a a a) (c)
  (a a a a a a) (b b b)
  (a a a a a a a) (d))
```

Under the hood, the cond^e here is allocating computational resource to four trivially different search space. The *unfair disj* in DFS_i allocates many more resources to the first search space. On the contrary, *fair disj* would allocate resources evenly to each search space.

```

142 ;; DFSf (fair disj)
143 > (run 12 q
144   (conde
145     ((repeato 'a q))
146     ((repeato 'b q))
147     ((repeato 'c q))
148     ((repeato 'd q))))
149 '((a) (b) (c) (d)
150   (a a) (b b) (c c) (d d)
151   (a a a) (b b b) (c c c) (d d d))
152
153 ;; BFS (fair disj)
154 > (run 12 q
155   (conde
156     ((repeato 'a q))
157     ((repeato 'b q))
158     ((repeato 'c q))
159     ((repeato 'd q))))
160 '((a) (b) (c) (d)
161   (a a) (b b) (c c) (d d)
162   (a a a) (b b b) (c c c) (d d d))

```

Running the same program again with almost-fair disj (e.g. DFS_{bi}) gives the same result. Almost-fair, however, is not completely fair, as shown by the following example.

```

156 ;; biDFS (almost-fair disj)
157 > (run 16 q
158   (conde
159     ((repeato 'a q))
160     ((repeato 'b q))
161     ((repeato 'c q))
162     ((repeato 'd q))
163     ((repeato 'e q))))
164 '((b) (c) (d) (a)
165   (b b) (c c) (d d) (e)
166   (b b b) (c c c) (d d d) (a a)
167   (b b b b) (c c c c) (d d d d) (e e))

```

DFS_{bi} is fair only when the number of goals is a power of 2, otherwise, some goals would be allocated twice as many resources than the others. In the above example, where the cond^e has five clauses, the clauses of b, c, and d are allocated more resources.

We end this subsection with precise definitions of all levels of disj fairness. Our definition of *fair disj* is slightly generalize from the one by Seres et al. [5]. Their definition is for binary disjunction. We generalize it to multi-arity one.

DEFINITION 2.1 (FAIR disj). A disj is fair if and only if it allocates computational resource evenly to search spaces produced by goals in the same disjunction (i.e. clauses in the same cond^e).

DEFINITION 2.2 (ALMOST-FAIR disj). A disj is almost-fair if and only if it allocates computational resource so evenly to search spaces produced by goals in the same disjunction that the maximal ratio of resources is bounded by a constant.

DEFINITION 2.3 (UNFAIR disj). A disj is unfair if and only if it is not even almost-fair.

2.2 fair conj

c1

^{c1}LKC: Checked grammar of text before this line

Given the following program, it is natural to expect lists of each letter to constitute 1/4 in the answer. Search strategies with unfair conj (e.g. DFS_i, DFS_{bi}, DFS_f), however, results in many more lists of a than lists of other letters. And some letters are rarely seen. The situation would be exacerbated if cond^e contains more clauses.

| | | |
|--|--|---|
| <pre>;; DFSi (unfair conj) > (run 12 q (fresh (x) (cond^e ((= 'a x)) ((= 'b x)) ((= 'c x)) ((= 'd x))) (repeat^o x q))) '((a) (a a) (b) (a a a) (a a a a) (b b) (a a a a a) (c) (a a a a a a) (b b b) (a a a a a a a) (d))</pre> | <pre>;; DFSf (unfair conj) > (run 12 q (fresh (x) (cond^e ((= 'a x)) ((= 'b x)) ((= 'c x)) ((= 'd x))) (repeat^o x q))) '((a) (a a) (b) (a a a) (a a a a) (b b) (a a a a a) (c) (a a a a a a) (b b b) (a a a a a a a) (d))</pre> | <pre>;; DFSbi (unfair conj) > (run 12 q (fresh (x) (cond^e ((= 'a x)) ((= 'b x)) ((= 'c x)) ((= 'd x))) (repeat^o x q))) '((a) (a a) (c) (a a a) (a a a a) (c c) (a a a a a) (b) (a a a a a a) (c c c) (a a a a a a a) (d))</pre> |
|--|--|---|

Under the hood, the cond^e and the call to repeat^o are connected by conj. The cond^e goal outputs a search space including four trivially different states. Then the next conjunctive goal, (repeat^o x q), is applied to each of these states, producing four trivially different search spaces. In the examples above, the conjs are allocating more computational resources to the search space of a. On the contrary, fair conj would allocate resources evenly to each search space. For example,

```
;; BFS (fair conj)
> (run 12 q
   (fresh (x)
    (conde
      ((= 'a x))
      ((= 'b x))
      ((= 'c x))
      ((= 'd x)))
    (repeato x q)))
'((a) (b) (c) (d)
  (a a) (b b) (c c) (d d)
  (a a a) (b b b) (c c c) (d d d))
```

A more interesting situation is when the first conjunct produces infinite many states. Consider the following example, a naive specification of fair conj might require search strategies to produce all sorts of singleton lists, but no longer ones, which makes the strategies incomplete. A search strategy is *complete* if and only if it can find out all the answers within finite time, otherwise, it is *incomplete*.

^{c2}LKC: Should I add a footnote?

“Although DFS_i’s di sj is unfair in general, it is fair when there is no call to relational definition in cond^e clauses (including this case).”

```

236      ;; naively fair conj
237      > (run 6 q
238         (fresh (xs)
239            (conde
240               ((repeat0 'a xs))
241               ((repeat0 'b xs)))
242            (repeat0 xs q)))
243      '(((a)) ((b))
244         ((a a)) ((b b))
245         ((a a a)) ((b b b)))

```

Our solution requires a search strategy with *fair conj* to organize states in bags in search spaces, where each bag contains finite states, and to allocate resources evenly among search spaces derived from states in the same bag. It is up to a search strategy designer to decide by what criteria to put states in the same bag, and how to allocate resources among search spaces related to different bags.

BFS puts states of the same cost in the same bag, and allocate resources carefully among search spaces related to different bags such that answers are produced in increasing order of cost. The *cost* of a answer is its depth in the search tree (i.e. the number of calls to relational definitions required to find them) [5]. In the following example, every answer is a list of list of symbols, where inner lists are the same. Here the cost of each answer is equal to the length of its inner lists plus the length of its outer list.

```

257      ;; BFS (fair conj)
258      > (run 12 q
259         (fresh (xs)
260            (conde
261               ((repeat0 'a xs))
262               ((repeat0 'b xs)))
263            (repeat0 xs q)))
264      '(((a)) ((b))
265         ((a) (a)) ((b) (b))
266         ((a a)) ((b b))
267         ((a) (a) (a)) ((b) (b) (b))
268         ((a a) (a a)) ((b b) (b b))
269         ((a a a)) ((b b b)))

```

We end this subsection with precise definitions of all levels of conj fairness.

DEFINITION 2.4 (FAIR conj). *A conj is fair if and only if it allocates computational resource evenly to search spaces produced from states in the same bag. A bag is a finite collection of state. And search strategies with fair conj should represent search spaces with possibly infinite collection of bags.*

DEFINITION 2.5 (UNFAIR conj). *A conj is unfair if and only if it is not fair.*

3 INTERLEAVING DEPTH-FIRST SEARCH

In this section, we review the implementation of interleaving depth-first search (DFS_i). We focus on parts that are relevant to this paper. TRS2 [2] provides a comprehensive description of the whole miniKanren implementation.

```

283 (define-syntax conde
284   (syntax-rules ()
285     ((conde (g ...) ...)
286      (disj (conj g ...) ...))))
287

```

Fig. 1. implementation of cond^e

The definition of cond^e (Fig. 1) is shared by all search strategies. It relates clauses disjunctively, and goals in the same clause conjunctively.

Fig. 2 shows parts that are later compared with other search strategies. We follow some conventions to name variables: variables bound to states are named *s*; variables bound to goals are named *g* (possibly with subscript); variables bound to search spaces have names ending with ∞ . The first function, *disj2*, implements binary disjunction. *if* and *only if* is its helper, which composes two disjunctive search spaces. The following function, *conj2*, implements binary conjunction. It applies the *first* goal to the input state, then applies the second goal to states in the resulting search space. The latter process is done with a helper function. *append-map* ∞ applies its input goal to states in its input search spaces and compose the resulting search spaces. It reuses *append* ∞ for search space composition. The following and last two definitions say disjunctive operator and conjunctive operator are right-associative.

4 BALANCED INTERLEAVING DEPTH-FIRST SEARCH

Balanced interleaving DFS (DFS_{bi}) has almost-fair *disj* and unfair *conj*. The implementation of DFS_{bi} differs from DFS_i's in the *disj* macro. We list the new *disj* with its helper in Fig. 3. When there are one or more disjunctive goals, *disj* builds a balanced binary tree whose leaves are the goals and whose nodes are *disj2*s, hence the name of this search strategy. The first “argument” to *disj+* is the goals. And the next two “arguments” accumulate goals to be put in the left and right subtrees. When there is only one goal, the tree is the goal itself. This case is handled in the first clause. When there are more goals, we partition the list of goals into two sublists of roughly equal lengths and recur on the two sublists. The partition is done by repetitively dispatching the first two goals in the list to the accumulators, until only zero or one goal remains. The two base cases of the partition are handled by the second and the third clauses respectively. And the dispatching is done at the fourth and last clause. In contrast, the *disj* in DFS_i constructs the binary tree in a particularly unbalanced form.

5 FAIR DEPTH-FIRST SEARCH

Fair DFS (DFS_f) has fair *disj* and unfair *conj*. The implementation of DFS_f differs from DFS_i's in *disj2* (Fig. 4). *disj2* is changed to call a new and fair version of *append* ∞ . *append* ∞_{fair} immediately calls its helper, *loop*, with the first argument, *s?*, set to #*t*, which indicates that *s* ∞ and *t* ∞ haven't been swapped. The swapping happens at the third *cond* clause in the helper, where *s?* is updated accordingly. The first two *cond* clauses essentially copy the cars and stop recursion when one of the input spaces is obviously finite. The third clause, as we mentioned above, is just for swapping. When the fourth and last clause runs, we know that both *s* ∞ and *t* ∞ are ended with a thunk and they have been swapped. In this case, a new thunk is constructed. Here changing the order of *t* ∞ and *s* ∞ won't hurt the fairness (though it will change the order of answers). We swap them back so that answers are produced in a more natural order.

```

330 #| Goal × Goal → Goal |#
331 (define (disj2 g1 g2)
332   (lambda (s)
333     (append∞ (g1 s) (g2 s))))
334
335 #| Space × Space → Space |#
336 (define (append∞ s∞ t∞)
337   (cond
338     ((null? s∞) t∞)
339     ((pair? s∞)
340      (cons (car s∞)
341            (append∞ (cdr s∞) t∞)))
342     (else (lambda ()
343              (append∞ t∞ (s∞))))))
344
345
346 #| Goal × Goal → Goal |#
347 (define (conj2 g1 g2)
348   (lambda (s)
349     (append-map∞ g2 (g1 s))))
350
351
352 #| Goal × Space → Space |#
353 (define (append-map∞ g s∞)
354   (cond
355     ((null? s∞) '())
356     ((pair? s∞)
357      (append∞ (g (car s∞))
358                (append-map∞ g (cdr s∞))))
359     (else (lambda ()
360              (append-map∞ g (s∞))))))
361
362
363 (define-syntax disj
364   (syntax-rules ()
365     ((disj) fail)
366     ((disj g) g)
367     ((disj g0 g ...) (disj2 g0 (disj g ...))))))
368
369 (define-syntax conj
370   (syntax-rules ()
371     ((conj) succeed)
372     ((conj g) g)
373     ((conj g0 g ...) (conj2 g0 (conj g ...))))))
374
375
376

```

Fig. 2. implementation of DFS_i


```

377 (define-syntax disj
378   (syntax-rules ()
379     ((disj) fail)
380     ((disj g ...) (disj+ (g ...) () ())))))
381
382 (define-syntax disj+
383   (syntax-rules ()
384     ((disj+ (g) () ()) g)
385     ((disj+ () (g1 ...) (gr ...))
386      (disj2 (disj+ (g1 ...) () ())
387              (disj+ (gr ...) () ())))
389     ((disj+ (g0) (g1 ...) (gr ...))
390      (disj2 (disj+ (g1 ... g0) () ())
391              (disj+ (gr ...) () ())))
392     ((disj+ (g0 g1 g ...) (g1 ...) (gr ...))
393      (disj+ (g ...) (g1 ... g0) (gr ... g1))))))
394
395
396
397
398
399
400
401
402
403
404
405
406
407
408
409
410
411
412
413
414
415
416
417
418
419
420
421
422
423

```

Fig. 3. implementation of DFS_{bi}

```

398 #| Goal  $\times$  Goal  $\rightarrow$  Goal |#
399 (define (disj2 g1 g2)
400   (lambda (s)
401     (append $\infty$ fair (g1 s) (g2 s))))
402
403 #| Space  $\times$  Space  $\rightarrow$  Space |#
404 (define (append $\infty$ fair s $\infty$  t $\infty$ )
405   (let loop ((s? #t)
406             (s $\infty$  s $\infty$ )
407             (t $\infty$  t $\infty$ ))
408     (cond
409      ((pair? s $\infty$ )
410       (cons (car s $\infty$ )
411             (loop s? (cdr s $\infty$ ) t $\infty$ )))
412      ((null? s $\infty$ ) t $\infty$ )
413      (s? (loop #f t $\infty$  s $\infty$ ))
414      (else (lambda ()
415              (loop #t (t $\infty$ ) (s $\infty$ )))))))
416
417
418
419
420
421
422
423

```

Fig. 4. implementation of DFS_f

6 BREADTH-FIRST SEARCH

BFS has both fair disj and fair conj . Our implementation is based on DFS_f (not DFS_i). To implement BFS based on DFS_f , we need $\text{append-map}_{\text{fair}}^\infty$ in addition to $\text{append}_{\text{fair}}^\infty$. The only difference between append-map^∞ and $\text{append-map}_{\text{fair}}^\infty$ is that the latter calls $\text{append}_{\text{fair}}^\infty$ instead of append^∞ .

The implementation can be improved in two ways. First, as mentioned in section 2.2, BFS puts answers in bags and answers of the same cost are in the same bag. In this implementation, however, it is unclear where this information is recorded. Second, $\text{append}_{\text{fair}}^\infty$ is extravagant in memory usage. It makes $O(n + m)$ new cons cells every time, where n and m are the “length”s of input search spaces. We address these issues in the first subsection.

Both our BFS and Seres’s BFS Seres et al. [5] produce answers in increasing order of cost. So it is interesting to see if they are equivalent. We prove so in Coq. The details are in the second subsection.

6.1 optimized BFS

As mentioned in section 2.2, BFS puts answers in bags and answers of the same cost are in the same bag. The cost information is recorded subtly – the cars of a search space have cost 0 (i.e. they are in the same bag), and the costs of answers in thunk are computed recursively then increased by one. It is even more subtle that $\text{append}_{\text{fair}}^\infty$ and the $\text{append-map}_{\text{fair}}^\infty$ respects the cost information. We make these facts more obvious by changing the type of search space, modifying related function definitions, and introducing a few more functions.

The new type is a pair whose car is a list of answers (the bag), and whose cdr is either a $\#f$ or a thunk returning a search space. A falsy cdr means the search space is obviously finite.

Functions related to the pure subset are listed in Fig. 5 (the others in Fig. 6). They are compared with Seres et al.’s implementation in our proof. The first three functions in Fig. 5 are search space constructors. `none` makes an empty search space; `unit` makes a space from one answer; and `step` makes a space from a thunk. The remaining functions do the same thing as before.

Luckily, the change in $\text{append}_{\text{fair}}^\infty$ also fixes the miserable space extravagance – the use of `append` helps us to reuse the first bag of t^∞ .

Kiselyov et al. [3] has shown that a *MonadPlus* hides in implementations of logic programming system. Our BFS implementation is not an exception: $\text{append-map}_{\text{fair}}^\infty$ is like `bind`, but takes arguments in reversed order; `none`, `unit`, and $\text{append}_{\text{fair}}^\infty$ correspond to `mzero`, `unit`, and `mplus` respectively.

Functions implementing impure features are in Fig. 6. The first function, `elim`, takes a space s^∞ and two continuations `ks` and `kf`. When s^∞ contains some answers, `ks` is called with the first answer and the rest space. Otherwise, `kf` is called with no argument. Here ‘s’ and ‘f’ means ‘succeed’ and ‘fail’ respectively. This function is like an eliminator of search space, hence the name. The remaining functions do the same thing as before.

6.2 comparison with the BFS of Seres et al. [5]

In this section, we compare the pure subset of our optimized BFS with the BFS found in Seres et al. [5]. We focus on the pure subset because their system is pure. Their system represents search spaces with streams of lists of answers, where each list is a bag.

To compare efficiency, we translate their Haskell code into Racket (See supplements for the translated code). The translation is direct due to the similarity in both logic programming systems and search space representations. The translated code is longer and slower than our BFS implementation. Details about difference in efficiency are in section 6.

We prove in Coq that the two BFSs are equivalent, i.e. $(\text{run } n \text{ } g)$ produces the same result (See supplements for the formal proof).

```

471 #|  $\rightarrow \text{Space}$  |#
472 (define (none) `(() . #f))
473
474 #|  $\text{State} \rightarrow \text{Space}$  |#
475 (define (unit s) `((,s) . #f))
476
477 #|  $(\rightarrow \text{Space}) \rightarrow \text{Space}$  |#
478 (define (step f) `(() . ,f))
479
480
481 #|  $\text{Space} \times \text{Space} \rightarrow \text{Space}$  |#
482 (define (append∞fair s∞ t∞)
483   (cons (append (car s∞) (car t∞))
484         (let ((t1 (cdr s∞))
485               (t2 (cdr t∞)))
486           (cond
487             ((not t1) t2)
488             ((not t2) t1)
489             (else (lambda () (append∞fair (t1) (t2))))))))
490
491 #|  $\text{Goal} \times \text{Space} \rightarrow \text{Space}$  |#
492 (define (append-map∞fair g s∞)
493   (foldr
494     (lambda (s t∞)
495       (append∞fair (g s) t∞))
496     (let ((f (cdr s∞)))
497       (step (and f (lambda () (append-map∞fair g (f))))))
498     (car s∞))
499
500
501 #|  $\text{Maybe Nat} \times \text{Space} \rightarrow [\text{State}]$  |#
502 (define (take∞ n s∞)
503   (let loop ((n n)
504             (vs (car s∞)))
505     (cond
506       ((and n (zero? n)) '())
507       ((pair? vs)
508        (cons (car vs)
509              (loop (and n (sub1 n)) (cdr vs))))
510       (else
511        (let ((f (cdr s∞)))
512          (if f (take∞ n (f)) '()))))))
513
514
515
516
517

```

Fig. 5. new and changed functions in optimized BFS that implements pure features

```

518 #| Space × (State × Space → Space) × (→ Space) → Space |#
519 (define (elim s∞ ks kf)
520   (let ((ss (car s∞))
521         (f (cdr s∞)))
522     (cond
523       ((and (null? ss) f)
524        (step (lambda () (elim (f) ks kf))))
525       ((null? ss) (kf))
526       (else (ks (car ss) (cons (cdr ss) f))))))
527
528
529 #| Goal × Goal × Goal → Goal |#
530 (define (ifte g1 g2 g3)
531   (lambda (s)
532     (elim (g1 s)
533           (lambda (s0 s∞)
534             (append-map∞fair g2
535              (append∞fair (unit s0) s∞)))
536           (lambda () (g3 s)))))
537
538
539 #| Goal → Goal |#
540 (define (once g)
541   (lambda (s)
542     (elim (g s)
543           (lambda (s0 s∞) (unit s0))
544           (lambda () (none)))))
545
546

```

Fig. 6. new and changed functions in optimized BFS that implements impure features

7 QUANTITATIVE EVALUATION

In this section, we compare the efficiency of search strategies. A concise description is in Table 1. A hyphen means running out of memory. The first two benchmarks are taken from Friedman et al. [2]. *reverso* is from Rozplokhas and Boulytchev [4]. Next two benchmarks about quine are modified from a similar test case in Byrd et al. [1]. The modifications are made to circumvent the need for symbolic constraints (e.g. \neq , absent^0). Our version generates de Bruijnized expressions and prevent closures getting into list. The two benchmarks differ in the cond^e clause order of their relational interpreters. The last two benchmarks are about synthesizing expressions that evaluate to '(I love you). This benchmark is also inspired by Byrd et al. [1]. Again, the sibling benchmarks differ in the cond^e clause order of their relational interpreters. The first one has elimination rules (i.e. application, car, and cdr) at the end, while the other has them at the beginning. We conjecture that DFS_i would perform badly in the second case because elimination rules complicate the problem when running backward. The evaluation supports our conjecture.

In general, only DFS_i and DFS_{bi} constantly perform well. DFS_f is just as efficient in all benchmarks but very-recursiveo. Both BFS have obvious overhead in many cases. Among the three variants of DFS (they all

| benchmark | size | DFS _i | DFS _{bi} | DFS _f | optimized BFS | Silvija's BFS |
|-----------------|--------|------------------|-------------------|------------------|---------------|---------------|
| very-recursiveo | 100000 | 579 | 793 | 2131 | 1438 | 3617 |
| | 200000 | 1283 | 1610 | 3602 | 2803 | 4212 |
| | 300000 | 2160 | 2836 | - | 6137 | - |
| appendo | 100 | 31 | 41 | 42 | 31 | 68 |
| | 200 | 224 | 222 | 221 | 226 | 218 |
| | 300 | 617 | 634 | 593 | 631 | 622 |
| reverso | 10 | 5 | 3 | 3 | 38 | 85 |
| | 20 | 107 | 98 | 51 | 4862 | 5844 |
| | 30 | 446 | 442 | 485 | 123288 | 132159 |
| quine-1 | 1 | 71 | 44 | 69 | - | - |
| | 2 | 127 | 142 | 95 | - | - |
| | 3 | 114 | 114 | 93 | - | - |
| quine-2 | 1 | 147 | 112 | 56 | - | - |
| | 2 | 161 | 123 | 101 | - | - |
| | 3 | 289 | 189 | 104 | - | - |
| '(I love you)-1 | 99 | 56 | 15 | 22 | 74 | 165 |
| | 198 | 53 | 72 | 55 | 47 | 74 |
| | 297 | 72 | 90 | 44 | 181 | 365 |
| '(I love you)-2 | 99 | 242 | 61 | 16 | 66 | 99 |
| | 198 | 445 | 110 | 60 | 42 | 64 |
| | 297 | 476 | 146 | 49 | 186 | 322 |

Table 1. The results of a quantitative evaluation: running times of benchmarks in milliseconds

have unfair conj), DFS_f is most resistant to clause permutation, followed by DFS_{bi} then DFS_i. Among the two implementation of BFS, ours constantly performs as well or better. Interestingly, every strategies with fair disj suffers in very-recursiveo and DFS_f performs well elsewhere. Therefore, this benchmark might be a special case. Fair conj imposes considerable overhead constantly except in appendo. The reason might be that strategies with fair conj tend to keep more intermediate answers in the memory.

8 RELATED WORKS

Edward points out a disjunct complex would be 'fair' if it is a full and balanced tree Yang [6].

Silvija et al Seres et al. [5] also describe a breadth-first search strategy. We proof their BFS is equivalent to ours. But our code is shorter and performs better in comparison with a straightforward translation of their Haskell code.

9 CONCLUSION

We analysis the definitions of fair disj and fair conj, then propose a new definition of fair conj. Our definition is orthogonal with completeness.

We devise two new search strategies (i.e. balanced interleaving DFS (DFS_{bi}) and fair DFS (DFS_f)) and devise a new implementation of BFS. These strategies have different features in fairness: _{bi} has almost-fair disj and unfair conj. DFS_f has fair disj and unfair conj. BFS has both fair disj and fair conj.

Our quantitative evaluation shows that DFS_{bi} and DFS_f are competitive alternatives to DFS_i, the current search strategy, and that BFS is less practical.

We prove our BFS is equivalent to the BFS in Seres et al. [5]. Our code is shorter and runs faster than a direct translation of their Haskell code.

ACKNOWLEDGMENTS

REFERENCES

- [1] William E Byrd, Michael Ballantyne, Gregory Rosenblatt, and Matthew Might. 2017. A unified approach to solving seven programming problems (functional pearl). *Proceedings of the ACM on Programming Languages* 1, ICFP (2017).
- [2] Daniel P. Friedman, William E. Byrd, Oleg Kiselyov, and Jason Hemann. 2018. *The Reasoned Schemer, Second Edition*.
- [3] Oleg Kiselyov, Chung-chieh Shan, Daniel P Friedman, and Amr Sabry. 2005. Backtracking, interleaving, and terminating monad transformers:(functional pearl). *ACM SIGPLAN Notices* 40, 9 (2005), 192–203.
- [4] Dmitri Rozplokhas and Dmitri Boulytchev. 2018. Improving Refutational Completeness of Relational Search via Divergence Test. In *Proceedings of the 20th International Symposium on Principles and Practice of Declarative Programming*. ACM, 18.
- [5] Silvija Seres, J Michael Spivey, and C. A. R. Hoare. 1999. Algebra of Logic Programming.. In *ICLP*. 184–199.
- [6] Edward Z. Yang. 2010. Adventures in Three Monads. *The Monad. Reader Issue* 15 (2010), 11.